AmirHosein Sadeghimanesh
2020 February, modified 2021 August

This worksheet reproduced the CAD representation of the multistationarity region of the network in Section 5.2 of the paper. This CAD was already produced in the paper "Kac-Rice formulas and the number of solutions of parametrized systems of polynomial equations, Elisenda Feliu, AmirHosein Sadeghimanesh, 2020". The CAD representation is shown in Figure 8c.

```
> F: = [-k[1]*x[1]+k[4]*x[3]*x[5], k[1]*x[1]-k[2]*x[2]+k[5]*x[4]*x
  [5], k[2]*x[2]-k[3]*x[3]-k[4]*x[3]*x[5], k[3]*x[3]-k[5]*x[4]*x
  [5], -k[4]*x[3]*x[5]-k[5]*x[4]*x[5]+k[6]*x[6], k[4]*x[3]*x[5]+k
  [5]*x[4]*x[5]-k[6]*x[6]]:
  <seq(F[i], i = 1..6)>: # The steady state equations.
> FF := [F[1], F[2], F[3], x[1] + x[2] + x[3] + x[4] - T[1], F[5],
  x[5] + x[6] - T[2]]:
  <seq(FF[i], i = 1 .. 6)>; # The 4th and the 6th steady state
  equations are linearly redundant and therefore replaced by
  conservation laws.
```

$$\left[\begin{array}{c} k_4 x_3 x_5 - k_1 x_1 \\ k_5 x_4 x_5 + k_1 x_1 - k_2 x_2 \\ -k_4 x_3 x_5 + k_2 x_2 - k_3 x_3 \\ x_1 + x_2 + x_3 + x_4 - T_1 \\ -k_4 x_3 x_5 - k_5 x_4 x_5 + k_6 x_6 \\ x_5 + x_6 - T_2 \end{array}\right] \tag{1}$$

Doing some simplifications from the following paper to convert the system to a univariate single polynomial (but still with 8 parameters).
V. B. Kothamachu, E. Feliu, L. Cardelli, and O. S. Soyer. Unlimited multistability and boolean logic in microbial signalling. J. R. Soc. Interface, 12(108):20150234, 2015.

```
> G := Groebner:-Basis(FF, plex(x[1], x[2], x[3], x[4], x[6], x[5])
  ):
  remove(i -> has(i, [x[1], x[2], x[3], x[4], x[6]]), G):
> f := eval(op(%),[k[1] = 0.7329000000, k[2] = 100, k[3] =
  73.2900000000, k[4] = 50, k[5] = 100, k[6] = 5, x[5] = t]);
```

$$\begin{aligned} f := {} & 2.5183225000 \times 10^6 t^3 + \big(366450.0000000000\, T_1 - 2.5183225000 \times 10^6\, T_2 \\ & + 63502.1205000000\big) t^2 + \big(537142.4100000000\, T_1 - 63502.1205000000\, T_2 \\ & + 26857.1205000000\big) t - 26857.1205000000\, T_2 \end{aligned} \tag{2}$$

```
> f := expand(25183225/10*t^3 + (366450*T[1] - 25183225/10*T[2] +
  635021205/10000)*t^2 + (53714241/100*T[1] - 635021205/10000*T[2]
  + 268571205/10000)*t - 268571205/10000*T[2]); # Remember that to
```

```
be used as an argument to the commands of RootFinding
[Parametric], not only the coefficients must be in rational type,
but also the polynomial should be in expanded shape.
```

$$f := \frac{5036645}{2} t^3 + 366450\, t^2 T_1 - \frac{5036645}{2} t^2 T_2 + \frac{127004241}{2000} t^2 + \frac{53714241}{100} t\, T_1 - \frac{127004241}{2000} t\, T_2 + \frac{53714241}{2000} t - \frac{53714241}{2000} T_2 \tag{3}$$

```
> eqCAD := [f = 0, 0 < t, 0 < T[1], 0 < T[2]]: # The system of
  equalitites and inequalities to be fed to the RootFinding
  [Parametric]'s CellDecomposition command.
> with(RootFinding[Parametric]):
> C := CellDecomposition(eqCAD, [t], [T[1], T[2]]):
> NumberOfSolutions(C); # Number of the open cells and the number
  of solutions in each of these cells.
```

$$[[1, 1], [2, 1], [3, 1], [4, 3], [5, 1], [6, 1]] \tag{4}$$

```
> CellPlot(C, 4, color = yellow, background = ColorTools:-Color(
  [0.57,0.88,1]), view = [0 .. 5, 0 .. 5], 'labels' = [k__7, k__8],
  'labelfont' = ["TimesNewRoman", 18], size=[460,460]); # Plotting
  the only open cell that has three solutions. This is the
  multistationarity region that we are looking for.
```



End of the file.